

Three Approaches to Estimate Power by SAS

Xiang Liu, Eli Lilly and Company, Shanghai, China

Abstract

To estimate power, there are different ways. This poster will introduce 3 approaches to estimate power by SAS. First of all, also the most common one is to use one SAS procedure – Proc Power. Secondly, given the power estimation formula, translating the formula to SAS code is a good option. However, in situation that lacking support of SAS procedure, and with estimation formula unknown, a solution of simulation is optimum. Examples will be utilized to illustrate the 3 ways.

Introduction

To demonstrate, an example of a randomized parallel balanced (1:1) non-inferiority trial will be adopted. The sample size calculation is based on a one-sided significance level of 0.025, a non-inferiority margin of 0.4%, and a standard deviation (SD) of 1.3% for the target study effect. As a result, with respect to 90% power, 223 patients per treatment arm would be needed, excluding the drop-out ones.

ESTIMATE OF POWER (by Proc Power)

First of all, the power estimate of the Proc Power is demonstrated as follows:

```
proc power;
  twosamplemeans
  test = diff
  sides = 1
  nulldiff = 0.4
  meandiff = 0
  stddev = 1.3
  alpha = 0.025
  npergroup = 223
  power = .
run;
```

Figure 1. Use Proc Power to estimate the power

```
The POWER Procedure
Two-sample t Test for Mean Difference
Fixed Scenario Elements

Distribution      Normal
Method           Exact
Number of Sides  1
Null Difference  0.4
Alpha            0.025
Mean Difference  0
Standard Deviation 1.3
Sample Size Per Group 223

Computed Power
Power
0.900
```

Figure 2. Result of power estimate from Proc Power

As you can see, it's easy to obtain the power, just provide the necessary parameters (expected sample size, type I error, etc.), as long as there is a procedure supported by SAS.

ESTIMATE OF POWER (by Formula)

In case you want to estimate a power which has not been supported by SAS yet, i.e. there is no corresponding SAS procedure currently, while you know the estimation formula. Refer to the previous example, the formula is as follows (lower one-sided):

$$\text{power} = \begin{cases} P(F(1, N-2, \delta^2) \geq F_{1-\alpha}(1, N-2)), & \text{two-sided} \\ P(t(N-2, \delta) \geq t_{1-\alpha}(N-2)), & \text{upper one-sided} \\ P(t(N-2, \delta) \leq t_{\alpha}(N-2)), & \text{lower one-sided} \end{cases}$$

$$\delta = N^{\frac{1}{2}}(w_1 w_2)^{\frac{1}{2}} \left(\frac{\mu_{diff} - \mu_0}{\sigma} \right)$$

Figure 3. Exact Power Computation for Two-Sample t Test Assuming Equal Variances

The following SAS code will implement the above formula to obtain the exactly same results as the Proc Power:

```
data power;
  alpha = 0.025;
  npergroup = 223;
  ntotal = npergroup * 2;
  nulldiff = 0.4;
  meandiff = 0;
  stddev = 1.3;
  df = ntotal - 2;
  weight1 = 0.5;
  weight2 = 0.5;
  ncp = sqrt(ntotal) * sqrt(weight1 + weight2) * ((meandiff - nulldiff) / stddev);
  t = tinv(alpha, df);
  power = probt(t, df, ncp);
  put ncp = ;
  put t = ;
  put power = ;
run;
```

Figure 4. To Estimate Power by Formula Implemented by SAS Code

NCP=-3.249032628
T=-1.965321285
POWER=0.9000844648

Figure 5. Results of Power Estimation by Formula

ESTIMATE OF POWER (by Simulation)

Given that neither SAS supported procedure, nor estimation formula, how to estimate the power? In this case, simulation is a good option.

To simulate the previous power estimation, the following SAS statements compute a power estimate with 10,000 simulations.

```
%let nulldiff = 0.4;
%let meandiff = 0;
%let stddev = 1.3;
%let alpha = 0.025;
%let npergroup = 223;
%let nsim = 10000;

data simdata;
  call streaminit(123);
  do isim = 1 to &nsim;
    do i = 1 to &npergroup;
      group = 1;
      y = rand('normal', &meandiff, &stddev);
      output;
      group = 2;
      y = rand('normal', &>nulldiff, &stddev);
      output;
    end;
  end;
run;

ods listing close;
ods output tests = tests;
proc ttest data = simdata sides = 1;
  class group;
  var y;
  by isim;
run;
ods listing;

data tests;
  set tests;
  if lowercase(method) = "pooled";
  sigflg = probt < &alpha;
run;
```

```
ods select binomialprop;
proc freq data = tests;
  tables sigflg / binomial(level = '1');
run;
```

Figure 6. SAS Code to Simulate Power Estimation

```
Binomial Proportion
for SIGFLG = 1
-----
Proportion          0.8989
```

Figure 7. Result of the Simulation of Power Estimation

As Figure 7 shows, the simulation result power = 0.8989 approximates that (0.900) by Proc Power and power formula.

You can obtain a highly accurate power estimate by simulating the power empirically. You need to use this approach for analyses that are not supported directly in SAS/STAT tools and for which you lack a power formula. But the simulation approach is also a viable alternative to existing power approximations. A high number of simulations will yield a more accurate estimate than a non-exact power approximation.

Conclusions

SAS POWER procedure is powerful for power estimation, far more than what has been discussed here. However, there are indeed some real problems beyond the scope of the POWER procedure or other similar ones (e.g. GLMPOWER, SEQDESIGN). Then you can try the other 2 options, either to compose the SAS code from corresponding formula, or to perform the simulation.

References

SAS/STAT 9.22 User's Guide – The POWER procedure

Acknowledgments

I'm grateful to Lucia Han for valuable assistance in the preparation of this paper.

Contact Information

Xiang Liu
 Eli Lilly and Company
 16F, 1 Corporate Avenue
 No. 222, Hu Bin Road
 Shanghai 200021 P.R.China
 +86-021-23021104
liu_xiang@lilly.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.